
The paper presents an enhancement of xASP, a system that generates explanation graphs for Answer
Set Programming (ASP). Different from xASP, the new system, xASP2 , supports different clingo
constructs like the choice rules, the constraints, and the aggregates such as #sum, #min. This work
formalizes and presents an explainable artificial intelligence system for a broad fragment of ASP,
capable of shrinking as much as possible the set of assumptions and presenting explanations in terms
of directed acyclic graphs.

1 Introduction

Recently, many modern artificial intelligence systems are increasingly capable of tackling complex prob-
lems. However, their lack of transparency can create a new issue: users may not comprehend why a
solution was obtained, making it difficult to trust the results. Moreover, with the right to an explanation
law extensively discussed in the USA, EU, and UK, and partly enacted in some countries, explainable
artificial intelligence (XAI) has experienced a substantial increase in interest. Thus, the focus of this
paper is on the development of an XAI system for Answer Set Programming (ASP) [10, 13]. Answer
Set Programming (ASP) [10, 13] is a well-known paradigm for problem-solving using logic programs
under answer set semantics [7] in knowledge representation and reasoning (KR&R) and an extension
of Datalog with a strong connection with well-founded semantics [14]. A variety of applications such
as planning, diagnosis, etc, have been successfully implemented using it. In this paper, our goal is to
provide an answer to the question “given an answer set A of a program Π and an atom α , why an atom
α is true (or false) in A?”.

The emergence of XAI has brought significant attention from researchers in ASP community, result-
ing in numerous proposed systems aimed at addressing this issue such as xclingo [4], DiscASP [9], xASP
[19], exp(ASPc) [18]. However, these systems are incapable of handling one or more of the following
scenarios: (i) false atoms can be explained by the system, (ii) the ability to support certain advanced lan-
guage features. In this paper, we proposed an improvement system, called xASP2 , that takes inspiration
from the approach used in xASP [19] and [16]. xASP2 are able to substantially increase scalability and
breadth of supported language features while producing explanation graphs with more immediately and
consistently useful to users. The improvement presented here deals with two main issues in explaining
the assignment of α in A: (i) how to compute a minimum cardinality set of atoms that is assumed to be
false such that it is capable of explaining the assignment of α in A; and (ii) how to support sophisticated
linguistic constructs such as choice rules and aggregates, which can be involved to explain the falsity of
some atoms in easily understandable terms.

Our main contributions are the following:

• A notion of explanation in terms of directed acyclic graphs explains why an atom is (or is not) in
an answer set in terms of easy-to-understand inferences originating from a hopefully minimum set

of assumed false atoms (Section 3). Note that the explanation graph of an atom is restricted to the
atoms involved in the relevant rules for the explaining atoms.

• A proof of existence for the explanations according to the given definition that guarantees the
correctness of our implementation (Section 4).

• The implementation of an enhancement system, xASP2 , for producing explanations powered by
ASP and its empirical evaluation(Sections 5–6). xASP2 tackles logic programs with different
clingo constructs such as aggregates and constraints.

The supported fragment of ASP includes uninterpreted function symbols, common aggregation func-
tions, comparison expressions, strong negation, constraints, normal rules, and choice rules. Aggregates
are expected to be stratified, to not involve default negation, and to have a single atomic condition.
Choice rules are expected to be unconditional, or otherwise to have exactly one conditional atom with a
self-explanatory condition (as for example a range expression or an extensional predicate). Additionally,
to ease the presentation, in Section 2 we only consider sum aggregates, and completely omit uninter-
preted function symbols, comparison expressions, strong negation, and conditions in choice rules. To
the best of our knowledge, this is the first explanation generation system that supports different clingo
constructs such as aggregates and constraints.

2 Background

All sets and sequences considered in this paper are finite. Let P, C, V be fixed nonempty sets of predicate
names, constants and variables. Predicates are associated with an arity, a non-negative integer. A term
is any element in C∪V. An atom is of the form p(t), where p ∈ P, and t is a possibly empty sequence
of terms. A literal is an atom possibly preceded by the default negation symbol not; they are referred to
as positive and negative literals.

An aggregate is of the form
sum{ta, t ′ : p(t)}⊙ tg (1)

where ⊙ is a binary comparison operator, p ∈ P, t and t ′ are possibly empty sequences of terms, and ta
and tg are terms.

A choice is of the form
t1 ≤ {atoms} ≤ t2 (2)

where atoms is a possibly empty sequence of atoms, and t1, t2 are terms. Let ⊥ be syntactic sugar for
1≤ {} ≤ 1.

A rule is of the form
head← body (3)

where head is an atom or a choice, and body is a possibly empty sequence of literals and aggregates. For
a rule r, let H(r) denote the atom or choice in the head of r; let BΣ(r), B+(r) and B−(r) denote the sets of
aggregates, positive and negative literals in the body of r; let B(r) denote the set BΣ(r)∪B+(r)∪B−(r).

A variable X occurring in B+(r) is a global variable. Other variables occurring among the terms t
of some aggregate in BΣ(r) of the form (1) are local variables. And any other variable occurring in r is
an unsafe variable. A safe rule is a rule with no unsafe variables. A program Π is a set of safe rules.
Additionally, we assume that aggregates are stratified, that is, the dependency graph GΠ having a vertex
for each predicate occurring in Π and an edge pq whenever there is r ∈Π with p occurring in H(r) and
q occurring in B+(r) or BΣ(r) is acyclic.

Figure 1: The undirected graph used as running example. Source vertices in blue, sink vertex in red.
Example 1. Given a connected undirected graph G encoded by predicate edge/2, source and sink nodes
encoded by predicates source/1 and sink/1, the following program assigns a direction to each edge so
that source nodes can still reach all sink nodes:

1≤ {arc(X ,Y ); arc(Y,X)} ≤ 1← edge(X ,Y ) (4)

reach(X ,X)← source(X) (5)

reach(X ,Y )← reach(X ,Z), arc(Z,Y ) (6)

⊥← source(X), sink(Y ), not reach(X ,Y ) (7)

If failures on the reachability condition are permitted up to a given threshold encoded by predicate
threshold/1, the program comprising rules (4)–(6) and

fail(X ,Y )← source(X), sink(Y ), not reach(X ,Y ) (8)

⊥← threshold(T ),sum{1,X ,Y : fail(X ,Y )}> T (9)

can be used. Note that X and Y are local variables in rule (9), and all other variables are global. ■

A substitution σ is a partial function from variables to constants; the application of σ to an expres-
sion E is denoted by Eσ . Let instantiate(Π) be the program obtained from rules of Π by substituting
global variables with constants in C, in all possible ways; note that local variables are still present in
instantiate(Π). The Herbrand base of Π, denoted base(Π), is the set of ground atoms (i.e., atoms with
no variables) occurring in instantiate(Π).

Example 2. Let Πrun comprise rules (4)–(6), (8)–(9) and the facts (i.e., rules with an empty body)
edge(a,b), edge(a,d), edge(d,c), source(a), source(b), sink(c), and threshold(0) (see Figure 1). Hence,
instantiate
(Πrun) contains, among others, the rules

1≤ {arc(a,b); arc(b,a)} ≤ 1← edge(a,b)

⊥← threshold(0),sum{1,X ,Y : fail(X ,Y )}> 0

and base(Πrun) contains fail(a,c), fail(b,c), and so on. ■

A (two-valued) interpretation is a set of ground atoms. For a two-valued interpretation I, relation
I |= · is defined as follows: for a ground atom p(c), I |= p(c) if p(c)∈ I, and I |= not p(c) if p(c) /∈ I; for an
aggregate α of the form (1), the aggregate set of α w.r.t. I, denoted aggset(α, I), is {⟨ta, t ′⟩σ | p(t)σ ∈ I,
for some substitution σ}, and I |= α if (∑⟨ca,c′⟩∈aggset(α,I) ca)⊙ tg is a true expression over integers; for a
choice α of the form (2), I |= α if t1 ≤ |I∩atoms| ≤ t2 is a true expression over integers; for a rule r with
no global variables, I |= B(r) if I |= α for all α ∈ B(r), and I |= r if I |= H(r) whenever I |= B(r); for a
program Π, I |= Π if I |= r for all r ∈ instantiate(Π).

For a rule r of the form (3) and an interpretation I, let expand(r, I) be the set {p(c)← body | p(c) ∈ I
occurs in H(r)}. The reduct of Π w.r.t. I is the program comprising the expanded rules of instantiate(Π)
whose body is true w.r.t. I, that is, reduct(Π, I) :=

⋃
r∈instantiate(Pi), I|=B(r) expand(r, I). An answer set of

Π is an interpretation A such that A |= Π and no I ⊂ A satisfies I |= reduct(Π,A).

Example 3. The only answer set Arun of program Πrun contains, among others, the atoms arc(b,a),
arc(a,d), arc(d,c), no other instance of arc/2, and no instance of fail/2. Hence, Arun |= 1≤ {arc(a,b);
arc(b,a)} ≤ 1 and Arun ̸|= sum{1,X ,Y : fail(X ,Y )}> 0. ■

A three-valued interpretation is a pair (L,U), where L,U are sets of ground atoms such that L ⊆U ;
sets L and U , also denoted (L,U)1 and (L,U)2, are the lower and upper bounds on the true atoms, so
atoms in L are true, atoms in U \L are undefined, and all other atoms are false. The evaluation function
[[·]]UL associates literals and aggregates with a truth value among u, t and f as follows: [[α]]UL = u if α is a
literal whose atom is p(c) and p(c)∈U \L, or α is an aggregate of the form (1) and aggset(α,U \L) ̸= /0,
or α is a choice of the form (2) and (U \L)∩atoms ̸= /0; [[α]]UL = t if [[α]]UL ̸= u and L |= α; and [[α]]UL = f
if [[α]]UL ̸= u and L ̸|= α . The evaluation function extends to rule bodies as follows: [[B(r)]]UL = f if there
is α ∈ B(r) such that [[α]]UL = f; [[B(r)]]UL = t if [[α]]UL = t for all α ∈ B(r); otherwise [[B(r)]]UL = u.

Example 4. For α being sum{1,X ,Y : fail(X ,Y )}> 0, [[α]]
{fail(a,c)}
/0 = u, [[α]]

{fail(a,c)}
{fail(a,c)} = t, and [[α]] /0

/0 = f.
■

Mainstream ASP systems compute answer sets of a given program Π by applying several inference
rules on (a subset of) instantiate(Π), the most relevant ones for this work summarized below. Let (L,U)
be a three-valued interpretation, and p(c) be a ground atom such that [[p(c)]]UL = u. Atom p(c) in H(r)
is inferred true by support if [[B(r)]]UL = t. (Actually, if H(r) is a choice of the form (2), inference by
support additionally requires that |atoms∩U | = t1, that is, undefined atoms in atoms∩U are required
to reach the bound t1. Such extra condition is not relevant for our work, and will not be used, because
our explanations aim at associating true atoms with rules with true bodies.) Atom p(c) is inferred false
by lack of support if each rule r ∈ instantiate(Π) with p(c) occurring in H(r) is such that [[B(r)]]UL = f.
Atom p(c) is inferred false by a constraint-like rule r ∈ instantiate(Π) if p(c) ∈ B+(r), [[H(r)]]UL = f and
[[B(r)\{p(c)}]]UL = t. Atom p(c) is inferred false by a choice rule r ∈ instantiate(Π) if H(r) has the
form (2), p(c) ∈ atoms, |atoms∩L| ≥ t2 and [[B(r)]]UL = t. Atom p(c) is inferred false by well-founded
computation if it belongs to some unfounded set X for Π w.r.t. (L,U), that is, a set X such that for all
rules r ∈ instantiate(Π) at least one of the following conditions holds: (i) no atom from X occurs in
H(r); (ii) [[B(r)]]UL = f; (iii) B+(r)∩X ̸= /0.

Example 5. Given the program instantiate(Πrun), and the three-valued interpretation ( /0,base(Πrun)),
atom edge(a,a) is inferred false by lack of support, atom source(a) is inferred true by support, and the set
{edge(a,a), arc(a,a)} is unfounded. Given ({arc(d,c)},base(Πrun)\{reach(a,c)}), atom reach(a,d)
is inferred false by the constraint-like rule (6), and arc(c,d) is inferred false by the choice rule (4). ■

3 Explanations

Let Π be a program, and A be one of its answer sets. A well-founded derivation for Π w.r.t. A, denoted
wf (Π,A), is obtained from the interpretation ( /0,base(Π)) by iteratively (i) adding to its lower bound
atoms of A that are inferred true by support, and (ii) removing from its upper bound atoms belonging to
some unfounded set. Note that wf (Π,A) is computed as a preprocessing step.

Example 6. Given Πrun and Arun from Examples 2–3, the lower bound of wf (Πrun,Arun) contains head
atoms in Example 2, arc(b,a), arc(a,d), arc(d,c), reach(a,a), reach(b,b), reach(a,d), reach(a,c),
reach(b,a), reach(b,c), and reach(b,d). The upper bound additionally contains arc(a,b), arc(d,a),
arc(c,d), and several instances of reach/2 and fail/2. ■

An explaining derivation for Π and A from (L,U) is obtained by iteratively (i) adding to L atoms
of A that are inferred true by support, and (ii) removing from U atoms that are inferred false by lack of

Figure 2: Induced DAG on the vertices reachable from arc(a,b) for the minimal assumption set /0 for Πrun.
support, constraint-like rules and choice rules. An assumption set for Π and A is a set X ⊆ base(Π)\A
of ground atoms such that the explaining derivation for Π and A from ( /0,wf (Π,A)2 \X) terminates with
A (in words, A is reconstructed from the false atoms of the well-founded derivation extended with X).
Let AS(Π,A) be the set of assumption sets for Π and A. A minimal assumption set for Π, A and a ground
atom α is a set X ∈ AS(Π,A) such that X ′ ⊂ X implies X ′ /∈ AS(Π,A), and α ∈ X implies α ∈ X ′ for
all X ′ ∈ AS(Π,A). (In other words, we prefer assumption sets not including the atom to explain. When
all assumption sets include the atom to explain, we opt for the singleton comprising the atom to explain
alone.) Let MAS(Π,A,α) be the set of minimal assumption sets for Π, A and α .

Example 7. Set base(Πrun)\Arun is an assumption set for Πrun and its answer set Arun. It can be checked
that also /0∈ AS(Πrun,Arun,α), and it is indeed the only minimal assumption set in this case, for any atom
in base(Πrun). ■

Given an assumption set X and an explaining derivation from ( /0,wf (Π,A)2 \X), a directed acyclic
graph (DAG) can be obtained as follows: The vertices of the graph are the atoms in base(Π) and the
aggregates occurring in instantiate(Π). (The vertex p(c) is also referred to as not p(c).) Any aggregate of
the form (1) is linked to instances of p(t). Atoms inferred true by support due to a rule r ∈ instantiate(Π)
are linked to elements of B(r). Any atom α inferred false by lack of support is linked to an element of
B(r) that is inferred false before α , for each rule r ∈ instantiate(Π) such that α occurs in H(r). Any
atom α inferred false by a constraint-like rule r ∈ instantiate(Π) is linked to the atoms occurring in H(r)
and the elements of B(r)\{α}. Any atom α inferred false by a choice rule r ∈ instantiate(Π) is linked
to the atoms occurring in H(r) that are true in A, and to the elements of B(r). A portion of an example
DAG is reported in Figure 2.

4 Existence of Minimal Assumption Sets

This section is devoted to formally show that the existence of minimal assumption sets is guaranteed,
and so are DAGs as defined in Section 3.

Theorem 1 (Main Theorem). Let Π be a program, A one of its answer sets, and α a ground atom in
base(Π). Set MAS(Π,A,α) is nonempty.

To prove the above theorem, we introduce some additional notation and claims. Let Π be a program,
and (L,U) be a three-valued interpretation. We denote by Π,L,U ⊢ α the fact that α ∈ base(Π) is
inferred true by support, which is the case when [[α]]UL = u, and there is r ∈ instantiate(Π) such that α

occurs in H(r) and [[B(r)]]UL = t, as defined in Section 2. Similarly, we denote by Π,L,U ⊢ not α the
fact that α ∈ base(Π) is inferred false by lack of support, constraint-like rules and choice rules, which is
the case when [[α]]UL = u, and one of the following conditions holds: each rule r ∈ instantiate(Π) with α

occurring in H(r) is such that [[B(r)]]UL = f; there is r ∈ instantiate(Π) with α ∈ B+(r), [[H(r)]]UL = f and
[[B(r)\{α}]]UL = t; there is r ∈ instantiate(Π) with H(r) of the form (2), α ∈ atoms, |atoms∩L| ≥ t2 and
[[B(r)]]UL = t.

The explaining derivation operator DΠ is defined as

DΠ(L,U) := (L∪{α ∈ base(Π) |Π,L,U ⊢ α},
U \{α ∈ base(Π) |Π,L,U ⊢ not α}).

Let (L,U)⊑ (L′,U ′) denote the fact that L⊆ L′ ⊆U ′ ⊆U , i.e., everything that is true w.r.t. (L,U) is true
w.r.t. (L′,U ′), and everything that is false w.r.t. (L,U) is false w.r.t. (L′,U ′).

Lemma 1. Operator DΠ is monotonic w.r.t. ⊑ .

Proof. For (L,U) ⊑ (L′,U ′), we shall show that DΠ(L,U) ⊑ DΠ(L′,U ′) holds. For α ∈ DΠ(L,U)1 \L
such that α /∈ L′, we have Π,L,U ⊢ α , that is, there is r ∈ instantiate(Π) such that α occurs in H(r) and
[[B(r)]]UL = t. As (L,U) ⊑ (L′,U ′), we have that [[B(r)]]U

′

L′ = t, that is, Π,L′,U ′ ⊢ α holds, and therefore
α ∈ DΠ(L′,U ′)1 \L.

For α ∈U \DΠ(L,U)2 such that α ∈U ′, we have Π,L,U ⊢ not α , and therefore we have three cases:

1. Each rule r ∈ instantiate(Π) with α occurring in H(r) is such that [[B(r)]]UL = f. As (L,U) ⊑
(L′,U ′), [[B(r)]]U

′

L′ = f holds.

2. There is r ∈ instantiate(Π) with α ∈ B+(r), [[H(r)]]UL = f and [[B(r)\{α}]]UL = t. As (L,U) ⊑
(L′,U ′), [[H(r)]]U

′

L′ = f and [[B(r)\{α}]]U ′L′ = t.

3. There is r ∈ instantiate(Π) with H(r) of the form (2), α ∈ atoms, |atoms∩L| ≥ t2 and [[B(r)]]UL = t.
As (L,U)⊑ (L′,U ′), |atoms∩L′| ≥ t2 and [[B(r)]]U

′

L′ = t.

In any case, Π,L′,U ′ ⊢ α holds, and therefore α ∈U ′ \DΠ(L′,U ′)2.

Lemma 2. L⊆ A⊆U implies DΠ(L,U)1 ⊆ A⊆ DΠ(L,U)2.

Proof. For α ∈DΠ(L,U)1\L we have Π,L,U ⊢α , that is, there is r∈ instantiate(Π) such that [[B(r)]]UL =
t. Hence, A |= B(r), and therefore expand(r,A) ⊆ reduct(Π,A). In particular, α ← B(r) belongs to the
reduct, and therefore α ∈ A.

For α ∈U \DΠ(L,U)2 we have Π,L,U ⊢ not α and we have to show that α /∈ A. Three cases:

1. Each rule r ∈ instantiate(Π) with α occurring in H(r) is such that [[B(r)]]UL = f.

2. There is r ∈ instantiate(Π) with α ∈ B+(r), [[H(r)]]UL = f and [[B(r)\{α}]]UL = t.

3. There is r ∈ instantiate(Π) with H(r) of the form (2), α ∈ atoms, |atoms∩L| ≥ t2 and [[B(r)]]UL = t.

In the first case, A\{α} |= reduct(Π,A), and therefore A\{α}= A because A is an answer set of Π. In
the other two cases, α /∈ A because A |= Π by assumption.

The explaining derivation from (L,U) is obtained as the fix point of the sequence (L0,U0) := (L,U),
(Li+1,Ui+1) := DΠ(Li,Ui) for i≥ 0. Note that the fix point is reached in at most |base(Π)| steps because
of Lemma 1 and each application of DΠ reduces the undefined atoms (or is a fix point). Thus, the system
eventually terminates in at most |base(Π)| steps.

Lemma 3. For any answer set A of Π, set base(Π)\A is an assumption set for Π and A.

Proof. Let (L,U) be the explaining derivation from ( /0,base(Π)\A). Thanks to Lemma 2, it is sufficient
to show that p(c) ∈ A implies p(c) ∈ L. Let us consider a topological ordering C1, . . . ,Cn (n≥ 1) for the
strongly connected components of GΠ, and let p ∈Ci. We use induction on i. Since p(c) ∈ A, there must
be r ∈ reduct(Π,A) such that H(r) = p(c) and A |= B(r). Hence, [[B−(r)]]UL = t. Moreover, [[BΣ(r)]]UL = t,
either because i = 1 and BΣ(r) = /0, or because of the induction hypothesis. Therefore, to have α /∈ L, it
must be the case that [[B+(r)]]UL ̸= t for all such rules, but in this case L |= reduct(Π,A), a contradiction
with the assumption that A is an answer set of Π.

Given Lemma 3, the proof of Main Theorem is immediate by the definition of MAS(Π,A,α) as
following:

Proof of Main Theorem. By definition, a minimal assumption set for Π, A and α is a set X ∈ AS(Π,A)
such that X ′ ⊂ X implies X ′ /∈ AS(Π,A), and α ∈ X implies α ∈ X ′ for all X ′ ∈ AS(Π,A). Lemma 3
guarantees the existence of an assumption set for Π and A. Existence of a minimal assumption set for Π,
A and α is therefore guaranteed.

5 Generation via Meta-Programming

By leveraging ASP systems, the concepts introduced in Section 3 can be computed. A meta-programming
approach is presented in this section, where the full language of ASP is used, including constructs omitted
in the previous sections, like weak constraints, uninterpreted functions, conditional literals and @-terms.
The reader is referred to [5] for details. We will use the name ASP programs for encodings using the full
language of ASP, in contrast to the name program that we use for encodings using the restricted syntax
introduced in Section 2.

Program Π, answer set A and the atom to explain are encoded by a set of facts obtained by com-
puting the unique answer set of the ASP program serialize(Π,A,α), defined next. Each atom p(c)
in base(Π) is encoded by a fact atom(p(c)); moreover, the encoding includes a fact true(p(c))
if p(c) ∈ A, and false(p(c)) otherwise; additionally, if p(c) is false in wf (Π,A), the encoding in-
cludes a fact explained_by(p(c), initial_well_founded). As for α , the encoding includes a
fact explain(α). Each rule r of instantiate(Π) is encoded by

rule(id(X)) :- atom(p1(t1)), ..., atom(pn(tn)).

where id is an identifier for r, X are the global variables of r, and B+(r)= {pi(ti) | i= 1, . . . ,n}; moreover,
the encoding includes

head(id(X),p(t)) :- rule(id(X)).

pos_body(id(X),p′(t ′)) :- rule(id(X)).

neg_body(id(X),p′′(t ′′)) :- rule(id(X)).

for each p(t) occurring in H(r), p′(t ′) ∈ B+(r) and p′′(t ′′) ∈ B(r); additionally, for each aggregate α of
the form (1) in BΣ(r), the encoding includes

pos_body(id(X),agg(X)) :- rule(id(X)).
aggregate(agg(X)) :- rule(id(X)).
true(agg(X)) :- rule(id(X)),

#sum{ta, t ′ : true(p(t))}⊙ tg.
false(agg(X)):- rule(id(X)), not true(agg(X)).

rule(agg(X)):- aggregate(agg(X)),true(agg(X)).

head(agg(X),agg(X)) :- rule(agg(X)).
pos_body(agg(X),p(t)):- rule(agg(X)),true(p(t)).
neg_body(agg(X),p(t)):-rule(agg(X)),false(p(t)).

rule((agg(X),p(t))) :- aggregate(agg(X)), false(agg(X)), atom(p(t)).
head((agg(X),p(t)),agg(X)):- rule((agg(X),p(t))).
pos_body((agg(X),p(t)),p(t)) :- rule((agg(X),p(t))), false(p(t)).
neg_body((agg(X),p(t)),p(t)) :- rule((agg(X),p(t))), true(p(t)).

where agg is an identifier for α; finally, if H(r) is a choice of the form (2), the encoding includes

choice(id(X),t1,t2) :- rule(id(X)).

Note that a true ground aggregate of the form (1) identified by agg(c) is associated with a single rule
whose body becomes true after all instances of p(t) are assigned the truth value they have in the answer
set A; on the other hand, a false aggregate is associated with one rule for each instance of p(t), whose
bodies becomes false when instances of p(t) are assigned the truth value they have in the answer set A.

Example 8. Recall Πrun and Arun from Examples 2–3. The ASP program serialize(Πrun,A,arc(a,b))
includes

atom(edge(a,b)). atom(arc(b,a)). atom (arc(a,b)). explain(arc(a,b)).
true(edge(a,b)). true(arc(b,a)). false(arc(a,b)).

rule(r4(X,Y)) :- atom(edge(X,Y)).
choice(r4(X,Y),1,1) :- rule(r4(X,Y)).
head(r4(X,Y), arc(X,Y)) :- rule(r4(X,Y)).
head(r4(X,Y), arc(Y,X)) :- rule(r4(X,Y)).
pos_body(r4(X,Y), edge(X,Y)):- rule(r4(X,Y)).

aggregate(agg1(T)) :- rule(r9(T)).
true(agg1(T)) :- rule(r9(T)), #sum{1,X,Y : true(fail(X,Y))} > T.

and several other rules. The answer set of serialize(Πrun,A,arc(a,b)) includes, among other atoms,
aggregate(agg1(0)) and false(agg1(0)). ■

The ASP program ΠMAS reported in Figure 3, coupled with a fact for each atom in the answer set of
serialize( ,A,α), has optimal answer sets corresponding to cardinality-minimal elements in MAS(Π,A,α).
Intuitively, line 1 guesses the assumption set, line 2–3 minimizes the size of the assumption set (prefer-
ring to not assume the falsity of the atom to explain), and lines 4–5 impose that each atom must have
exactly one explanation. The other rules encode the explaining derivation for Π and A from wf (Π,A)\X ,
where X is the guessed assumption set.

Given a minimal assumption set encoded by predicate assume_false/1, an explaining derivation
can be computed by removing lines 1–3 from the ASP program ΠMAS. Let ΠEXP be such an ASP program.
Finally, given an explaining derivation encoded by explained_by(Index,Atom,Reason), with the
additional Index argument encoding the order in the sequence, a DAG linking atoms according to the
derivation can be computed by the ASP program ΠDAG reported in Figure 4.

Example 9. Let ΠS have a fact for each atom in the answer set of serialize(Πrun,Arun,arc(a,b)). ΠMAS∪
ΠS generates the empty assumption set. ΠEXP∪ΠS ∪ /0 generates an explaining derivation, for example
one including explained_by(edge(a,b),(support,r6)), explained_by(arc(b,a),(support,r1(a,b))) and
explained_by(arc(a,b),(choice_rule,r1(a,b))). Let ΠE have a fact for each instance of explained_by/3

1 {assume_false(Atom)} :- false(Atom), not aggregate(Atom).
2 :∼ false(Atom), assume_false(Atom), not explain(Atom). [1@1, Atom]
3 :∼ false(Atom), assume_false(Atom), explain(Atom). [1@2, Atom]

4 has_explanation(Atom) :- explained_by(Atom,_).
5 :- atom(X), #count{Reason: explained_by(Atom,Reason)} != 1.

6 explained_by(Atom, assumption) :- assume_false(Atom).

7 {explained_by(Atom, (support, Rule))} :- head(Rule,Atom), true(Atom);
8 true (BAtom) : pos_body(Rule,BAtom); has_explanation(BAtom) : pos_body(Rule,BAtom);
9 false(BAtom) : neg_body(Rule,BAtom); has_explanation(BAtom) : neg_body(Rule,BAtom).

10 {explained_by(Atom, lack_of_support)} :- false(Atom); false_body(Rule) : head(Rule,Atom).
11 false_body(Rule) :- rule(Rule); pos_body(Rule,BAtom), false(BAtom), has_explanation(BAtom).
12 false_body(Rule) :- rule(Rule); neg_body(Rule,BAtom), true(BAtom), has_explanation(BAtom).

13 {explained_by(Atom, (required_to_falsify_body, Rule))} :- false(Atom), not aggregate(Atom);
14 pos_body(Rule,Atom), false_head(Rule); true(BAtom) : pos_body(Rule,BAtom), BAtom != Atom;
15 has_explanation(BAtom) : pos_body(Rule,BAtom), BAtom != Atom;
16 false(BAtom) : neg_body(Rule,BAtom); has_explanation(BAtom) : neg_body(Rule,BAtom).
17 explained_head(Rule) :- rule(Rule); has_explanation(HAtom) : head(Rule,HAtom).
18 false_head(Rule) :- explained_head(Rule), not choice(Rule,_,_);
19 false(HAtom) : head(Rule,HAtom).
20 false_head(Rule) :- explained_head(Rule), choice(Rule, LowerBound, UpperBound);
21 not LowerBound <= #count{HAtom' : head(Rule,HAtom'), true(HAtom')} <= UpperBound.

22 {explained_by(Atom, (choice_rule, Rule))} :- false(Atom);
23 head(Rule,Atom), choice(Rule, LowerBound, UpperBound);
24 true(BAtom) : pos_body(Rule,BAtom); has_explanation(BAtom) : pos_body(Rule,BAtom);
25 false(BAtom) : neg_body(Rule,BAtom); has_explanation(BAtom) : neg_body(Rule,BAtom);
26 #count{HAtom : head(Rule, HAtom), true(HAtom), has_explanation(HAtom)} = UpperBound.

Figure 3: ASP program ΠMAS for computing a minimal assumption set
in the explaining derivation. ΠDAG∪ΠS∪ΠE generates a DAG, for example one including link(arc(b,a),
edge(a,b)), link(arc(a,b),arc(b,a)) and link(arc(a,b),edge(a,b)). ■

6 Implementation and Experiment

We deployed an XAI system for ASP named xASP2 , which is powered by the clingo python api

[8]. By taking an ASP program Π, one of its answer sets A, and an atom α as input, xASP2 is capable
of producing minimal assumption sets, explaining derivations, and DAGs as output to assist the user in
determining the assignment of α . The source code is available at https://github.com/alviano/xasp and an
example DAG is given at https://xasp-navigator.netlify.app/.

The pipeline implemented by xASP2 starts with the serialization of the input data, which is obtained
by means of an ASP program crafted from the abstract syntax tree of Π and whose answer set identifies
the relevant portion of instantiate(Π) and base(Π). In a nutshell, ground atoms provided by the user,
A∪ {α}, are part of base(Π) and used to instantiate rules of Π (by matching positive body literals),
which in turn may extend base(Π) with other ground atoms occurring in the instantiated rules; possibly,
some atoms of base(Π) of particular interest can be explicitly provided by the user. Aggregates are also
processed automatically by means of an ASP program, and so is the computation of false atoms in the
well-founded derivation wf (Π,A).

Obtained serialize(Π,A,α), xASP2 proceeds essentially as described in Section 5, by computing
a minimal assumption set, an explaining derivation and an explanation DAG. As an additional opti-

https://github.com/alviano/xasp
https://xasp-navigator.netlify.app/

1 link(Atom, BAtom) :- explained_by(_, Atom, (support, Rule)); pos_body(Rule, BAtom).
2 link(Atom, BAtom) :- explained_by(_, Atom, (support, Rule)); neg_body(Rule, BAtom).

3 {link(Atom, A) : pos_body(Rule,A), false(A), explained_by(I,A,_), I < Index;
4 link(Atom, A) : neg_body(Rule,A), true (A), explained_by(I,A,_), I < Index} = 1 :-
5 explained_by(_, Atom, lack_of_support); head(Rule, Atom).

6 link(Atom,A) :- explained_by(_, Atom, (required_to_falsify_body, Rule)); head(Rule,A).
7 link(At,A) :- explained_by(_,At,(required_to_falsify_body, Rule)); pos_body(Rule,A), A!=At.
8 link(Atom,A) :- explained_by(_,Atom,(required_to_falsify_body, Rule)); neg_body(Rule,A).

9 link(Atom, HAtom) :- explained_by(_,Atom,(choice_rule, Rule)); head(Rule,HAtom), true(HAtom).
10 link(Atom, BAtom) :- explained_by(_, Atom, (choice_rule, Rule)); pos_body(Rule, BAtom).
11 link(Atom, BAtom) :- explained_by(_, Atom, (choice_rule, Rule)); neg_body(Rule, BAtom).

Figure 4: ASP program ΠDAG for computing a directed acyclic graph associated with an explaining derivation
mization, the explaining derivation is shrunk to the atoms reachable from α , utilizing an ASP pro-
gram. Finally, the user can opt for a few additional steps: obtain a graphical representation by means
of the igraph network analysis package (https://igraph.org/); obtain an interactive representation in
https://xasp-navigator.netlify.app/; ask for different minimal assumption sets, explaining derivations and
DAGs.

We assessed xASP2 empirically on the commercial application. The ASP program comprises 420
rules and 651 facts. After grounding, there are 4261 ground rules and 4468 ground atoms. The program
was expected to have a unique answer set, but two answer sets were actually computed. Our experiment
was run on an Intel Core i7-1165G7 @2.80 GHz and 16 GB of RAM. xASP2 computed a DAG for the
unexpected true atom, behaves_inertially(testing_posTestNeg,121), in 14.85 seconds on
average, over 10 executions. The DAG comprises 87 links, 45 internal nodes and 20 leaves, only one of
which is explained by assumption; only 30 of the 420 symbolic rules and 11 of the 651 facts are involved
in the DAG; at the ground level, only 48 of the 4261 ground rules and 65 of the 4468 ground atoms
are involved. Additionally, we repeated the experiment on 10 randomly selected atoms with respect to
two different answer sets, repeating each test case 10 times. We measured an average runtime of 14.79
seconds, with a variance of 0.004 seconds.

Table 1: The action preconditions and effects in Blockworlds problem

Action Precondition Effects
stack(X ,Y )
- stack block X is on
block Y

Block Y is clear
The agent holds the block X

X is clear
X is on Y
Y is no longer clear
The agent does not hold anything

unstack(X ,Y )
- unstack block X is
on block Y

X is clear
X is on Y
The agent does not hold anything

The agent holds the block X
Y becomes clear
X is not clear

pickup(X)
- pickup block X
from the table

X is clear
X is on the table
the agent does not hold anything

The agent holds the block X
X is no longer on the table and is not
clear

putdown(X)
- put down block X
onto the table

The agent holds the block X X is clear
X is on the table
the agent does not hold anything

xASP2 also has the ability to handle explainable planning, meaning it can generate an explanation

https://igraph.org/
https://xasp-navigator.netlify.app/

Figure 5: The initial and goal states of Blocksworld.

1 h(X,T+1) :- action(action(A)),occurs(A,T), postcondition(action(A),
effect(unconditional),X,value(X,true)).

2 -h(X,T+1) :- action(action(A)),occurs(A,T), postcondition(action(A),
effect(unconditional),X,value(X,false)).

3 h(X,T+1) :- h(X,T), not -h(X,T+1).
4 -h(X,T+1) :- -h(X,T), not h(X,T+1).
5 non_exec(A,T) :- action(action(A)), not h(X,T), precondition(action(A),X,value(X, true)).
6 non_exec(A,T) :- action(action(A)), not -h(X,T), precondition(action(A),X,value(X, false)).
7 :- action(action(A)),occurs(A,T), non_exec(A,T).

Figure 6: ASP program for reasoning about effects of actions [12]
graph showing why a particular action cannot take place at a certain time. To demonstrate this capability,
we will use a popular problem known as Blocksworld. The initial state (left) and goal state (right) of the
problem are shown in Figure 5. Five fluents are on(X ,Y ) - block X is on block Y , onTable(X) - block
X is on the table, clear(X) - block X is clear, holding(X) - the agent holds the block X , and handEmpty
- the agent does not hold anything. Four differnt actions are stack, unstack, pickup and putdown. The
domain description of the problem is shown in Table 1 in which the predictions and effects of four actions
are presented.

The rules for reasoning about effects of actions, action generation and goal enforcement [12] are
utilized as programming input in xASP2 . Figure 6 shows the ASP program for reasoning about the
effects of actions in which an action occurs only when its preconditions are true and then its effects are
true in the next time step. Specifically, lines 5 and 6 are used to define states in which an action cannot
be executed, and constraint is employed to prevent non-executable actions from occurring (line 7).

For the problem described in Figure 5, executing the actions of unstack(a,b), putdow(a), pickup(b)
and stack(b,a) at times 0, 1, 2, and 3 respectively constitutes the optimal plan (assuming time starts at
0). However, if users are in a rush and want to put down block a on the table at time 0, as represented
by the atom occurs((”putdown”,constant(”a”)),0), they will encounter a false occurrence of the ac-
tion putdown(a). Figure 6 shows that atom occurs((”putdown”,constant(”a”)),0) is false because the
constraint rule prevents its execution and the prediction of the action holding block a is invalid/false.

7 Related Work

Our work, as stated in the introduction, is situated within the realm of XAI and can be used as a debugging
tool to provide explanations for an unexpected result. For instance, if an element α is false in all answer
sets of a program Π despite user expectations, our system can help identify which rules are contributing
to this anomalous behavior. Thus, in this section, we will explore both debugging tools for ASP and
cutting-edge XAI systems designed for ASP.

In Table 2, a summary of the compared features is presented. The features include as follows:

Figure 7: The DAG for atom occurs((”putdown”,constant(”a”)),0).
whether the explanation is guaranteed to be acyclic; the capability to handle the input program with
aggregates and constraints; the ability to provide an explanation when the query atom can be false in the
answer set; and whether the system is available for experimentation. As can be seen from Table 2, our
system is capable of providing explanations for false atoms and does not lead to cyclic argumentation
in the explanation. xASP2 is the only system that tackles a program that includes both aggregates and
constraints. It is worth noting that the topic of aggregates is addressed in another approach [11], even
though no system implementing this approach is mentioned or available.

8 Conclusion

We have developed and implemented xASP2 , an XAI system that targets the ASP language and is
powered by ASP engines. Our approach to explaining why an atom is true/false in an answer set involves
deriving easy-to-understand inferences originating from a hopefully small set of atoms assumed false.
xASP2 has the ability to support different clingo constructs such as aggregates and constraints. It
produces an explanation as a DAG with the atom to be explained as the root and takes a few seconds
to compute the explanation in our test cases. Further investigation to include ASP’s other linguistic

Table 2: Summary of compared features

System (if any)
and reference

Acyclic
explanation

Linguistic
extentions

Explanation for
false atoms

System
availability

s(CASP) [1] Yes Constraints Yes Yes
ASPeRiX [2] Yes Constraints Yes Yes
spock [3] Yes Constraints No Yes

xclingo [4] Yes None No Yes
DWASP [6] Yes Constraints No Yes

[11] No Aggregates Yes No
Visual-DLV [15] Yes Constraints No Yes

[16] No None Yes No
LABAS [17] No None Yes Yes

exp(ASPc) [18] No Constraints Yes Yes
[20] Yes None Yes No

xASP2 Yes Aggregates and Constraints Yes Yes

constructs such as conditional literals, beyond those currently supported, present in the future work.

